

INTRODUCTION

This HP-19C/HP-29C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become an expert on your HP calculator.

You will find general information on how to key in and run programs under "A Word about Program Usage" in the Applications book you received with your calculator.

We hope that this Solutions book will be a valuable tool in your work and would appreciate your comments about it.

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

TABLE OF CONTENTS

RACETRACK	1
PINBALL MACHINE	6
15 BALL ROTATION POOL	, 10
ROULETTE	, 13
TIC-TAC-TOE	. 16
BELL FRUIT	. 20
BLACKJACK (21)	, 23
CANNIBALS & MISSIONARIES	. 26
HUNT A MOVING SUBMARINE	. 29
ARTILLERY GAME	, 33

This game simulates a two-car race (or a one-car race against time) on a track of arbitrary shape. The track shown may be used, or other tracks of any shape may be designed. The program computes the velocities and positions of the cars.

Initially, both cars are at rest on the start-finish line. Car 1 is at (x,y) position (0,0) and car 2 is at position (5,0.). Player 1 starts. For each move, a player may accelerate in any direction or coast. To make a move, enter the direction (in degrees) and magnitude (0-9) of acceleration and press car number (1 or 2). To coast, enter any direction and a magnitude of zero. For a panic stop, enter a direction exactly opposite $(180^{\circ} \text{ away})$ from your present direction and use a magnitude of 9. The faster you go, the more moves it will take to stop completely.

After making your move, the display will show your car's velocity.

By rolling down the stack, the car's direction of travel, and it's (x,y) position may be displayed.

Traction limits the maximum rate of change of velocity to 9 meters per second per second. Note that this realistically limits the ability to turn when accelerating or declerating.

If the center points of the two cars get within 2 meters of each other, the cars collide, and the display shows flashing zeros. This destroys the two cars, and the game must be started over from the beginning.

It is most convenient to use a fresh sheet of graph paper with the track drawn on it for each game; then the positions of the cars may be plotted, and sequential positions joined by straight line segments. The players must decide after each move whether the car is off of the track, or if it had to go off of the track to travel between the last two positions.

EQUATIONS:

$$V_{f} = V_{i} + \Delta V$$

 $P_{f} = P_{i} + ((V_{i} + V_{f})/2)t$

where

- ΔV = Velocity change due to 1 sec. accel.
- $P_i & P_f = Initial & final positions (x & y)$
 - t = Time, seconds (1 sec. in this prog.)

NOTES:

The program halts, displaying zero if $+ 90^{\circ}$ is used in P \rightarrow R function; press "R/S" to continue. Direction of car's travel is with respect to the fixed frame of reference of car's starting position (origin) and the start-finish line (0°). See the sketch.

All directions are entered and displayed as A + or - Angle between 0° and 180° .

A collision of the two cars (flashing zeros) ends the game.


Each move advances the car one second in time.

REFERENCES:

Martin Gardner, Mathematical Games, <u>Scientific</u> <u>American</u>, Jan. 1973 and May 1973.

REFERENCES: (continued)

This program is adapted from HP-65 Users' Library program #04326A by Delmer D. Hinrichs.




RULES:

- A car which goes off the track, either at a plotted position or between plotted positions, loses the race.
- 2. A car which collides with the other one loses.
- 3. If both cars cross the start/finish line on the same move, the car which finishes farthest from the line wins.

SOLUTIONS:

	GSBØ	
45.00	ENT↑	
9.00	GSB1	
9.00	***	Car l velocity
	R↓	
45.00	***	Direction
	R∔	
3.18	***	x-position
	R↓	·
3.18	***	y-position
45.00	FUTA	
43.00 9.00		
5.00 9.00		Car 2 velocity
3.00		
AE 00	R↓	Direction
45.00		
0 10	R↓	
8.18	***	x-position
7 10	R↓	y-position
3.18	***	y-position
60.00	ENT↑	
9.00	GSB1	
17.85	***	Car 1 velocity
	R∔	
52.50	***	Direction
	R↓	
11.80	***	x-position
	R↓	
13.44	***	y-position
135.00	FNT †	
9.00	GSB2	Car 2
0.00	***	(flashing)collision car 2 loses



RACETRACK

You may wish to copy this to play your game. For variety, you may wish to draw your own track.

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Initialize		GSB	0	5.00
3.	Enter parameters for car x*:				
	Direction of acceleration, degress	Angle	ENT↑		
	Magnitude of acceleration	Acc	GSB	n*	Velocity
	<pre>* n = 1 for car 1; n = 2 for car 2</pre>				
4.	Display car n's status:				
	Velocity, meters/sec				Velocity
	Direction of travel, degrees		R↓		Direction
	x-position, meters		R↓		x-pos.
	y-position, meters		R↓		y-pos.
5.	For next move, go to step 3.				
6.	For a new game, go to step 2				
7.	Flashing zeros indicate a collision and				
	the end of the game.				

		0	T		
01 *LBL0	Initia	lize	50 *LBL6		
02 FIX2			51 RCL		
03 DEG					
04 CLRG			52 +	New v	velocity, V _f velocity, V _f
05 5	Car 2	x position	53 STO:	01d v	velocity, V
06 ST06		··	54 LSTX		1
07 R/S			55 +		
			56 2		
08 *LBL1	Car 1		57 ÷		
09 1			58 ISZ		
10 STO0			59 ST+i	New	position, P _f
11 GT09			60 RTN		f f
12 *LBL2	Car 2		61 R/S		
13 5					
14 STO0					
15 *LBL9	Acc.				
16 R4					
17 9					
18 X <u>4</u> Y?		acc. to			
19 XZY	9 m/	sec ²			
20 R↓					
21 +R	y comp	onent y comp.			
21 7K 22 GSB6					
22 6366 23 X≠Y	y comp	ate V _f and P _f			
		ate V and P			
24 ISZ	carcur	ate V _f and P _f			
25 GSB6					
26 RCL8	1)				
27 RCL4					
28 -	Cha	ck for			
29 RCL6		ollision			
30 RCL2					
31 -	('•	e.<2m.)			
32 →P					
33 2					
34 X>Y?	J				
35 GT05	1				
36 RCLI					
37 DSZ					
38 RCL:					
39 DSZ					
40 RCL:	 Pre	epare display			
41 DSZ					
42 X≠Y					
43 RCL i					
44 →P	** * "P	rint Stack" ma	y be inserte	d before "R/S	
45 R/S	J		[
46 *LBL5					
47 0	Blink	ing zero			
48 PSE					
49 GT05					
10 0700		RFGI	LSTERS	I	
⁰ Pointer	¹ x ₁ vel	$ ^2 x_1 \text{ pos} $	$3 y_1 vel$	4 V. DOS	⁵ x ₂ vel
6		8	9 y 1 vei	⁴ y ₁ pos	.1
X ₂ pos	<u>y₂ vel</u>	y ₂ pos	.5	16	17
18	19	20	-		
	-		21	22	23
24	25	26	27	28	29

PINBALL MACHINE

This game allows multiple scoring for 5 balls. The balls may be controlled with 4 flippers and tilting is also allowed. As the ball scores, the score is flashed and the running total is displayed. When the ball falls, the display blinks zero. New balls may be set up and played until the 5th ball falls; at that time the total score for the game is displayed as a negative number.

NOTE:

On very rare occasions, the machine will overflow (θ =90°...). In that event, store some other seed in R₂ or start a new game.

This program is adapted from HP-65 Users' Library program #03458A by Peter C. Wang. SOLUTION: **GSB0** Initialize 0.32147 STO2 * GSB1 Use flipper [1] 200. *** Total score GSB3 [3] Flashing Ø-Ball | falls RCL0 4 balls left *** 4. GSB2 [2] 500. *** Score=300,tota1=500 GSB2 [2] 1800. *** Total gsb1 [1] Flashing Ø-Ball 2 falls RCLØ 3. *** 3 Balls Left GSB4 [4] 2100. *** GSB2 2700. *** GSB1 3300. *** GSB1 8000. *** GSB4 8800. *** GSB1 8900. *** GSB2 9400. *** GSB3 11300. *** To reproduce this GSB4 example, store 0.32147 12000. *** in R2. When playing GSB4 follow user 12800. *** instruction 2. GSB4 12800. *** GSB4 15500. *** GSB4 16600. *** GSB2 26200. ***

SOLUTION:

GSB3 26400. *** GSB1 27200. *** Score=800,Tota1=27,200 GSB3 27300. *** GSB1 27800. *** GSB4 28600. *** GSB3 29500. *** GSB1 Flashing Ø-Ball 3 falls RCLØ 2. *** 2 balls left GSB3 RCLØ 1. *** 1 ball left GSB2 29500. *** GSB3 30000. *** GSB1 30900. *** GSB4 31000. *** GSB1 32200. *** GSB1 32700. *** GSB3 39800. *** GSB2 39800. *** GSB2 40200. *** GSB3 40900. *** GSB2 -40900. *** Last ball falls, total= 40,900

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEY	/S	OUTPUT DATA/UNITS
1.	Key in the program				
2.	Initialize let initialization run		GSB	0	
	between 1-45 seconds; then halt by		Any		
	pressing any key		Кеу		
3.	Play the ball with any of the four		GSB	1	
	flippers. The score flashes and the total	or	GSB	2	
	is displayed	or	GSB	3	Score/total
		or	GSB	4	
4.	Repeat step 3 with the same ball until a				
	blinking zero appears, indicating the				
	ball has fallen.				
5.	When the ball has fallen, set up a new	Blinking	Any		
	ball and go to step 3 to play it.	zero	Кеу		0.
6.	The number of balls remaining may be		RCL	0	<pre># of balls</pre>
	recalled at any time				Left
7.	When the fifth ball has fallen the game				
	is over and the total score is displayed				
	as a negative number				
8a.	You may "tilt" the machine at any time		GSB	5	
	(this operation changes the seed)				
8b.	Stop the tilting operation by pressing		Any		
	any key. Go to step 3.		Кеу		
9.	For a new game, go to step 2.				

			T					٦
01 *LBL0 02 CLRG 03 1 04 ST02 05 . 06 9 07 8 08 5 09 3 10 ST03 11 EEX 12 2 13 ST05 14 8 15 5 16 ST07 17 5 18 ST00 19 FIX0 20 *LBL5 21 RCL3 22 ST×2 23 GT05 24 *LBL1 25 3 26 9 27 GT04 28 *LBL2 29 5 30 3 31 GT04 32 *LBL3 33 RCL7 34 *LBL4 35 RCL2 36 RCL7	tiltin (.98 n=#o 39=k 53=k 53=k 1astx fl (85) F	<pre>f balls=5 ng operation 53)ⁿ>R2 f loops for flipper[1] for flipper[2] for flipper[3] = k for ipper [4] R2 + K</pre>	67 68 69 70 71 72 73 74 75 76 77	ST02 RCL4 RCL7 X4Y? GT08 RCL2 1 0 X INT RCL5 X ST+1 PSE RCL1 R/S *LBL8 DSZ GT09 RCL1 CLRG CHS R/S * CHS R/S CHS R/S CHS R/S		O < t "round Σ scor Displa Reduce Indica ball Displa as a	alls(tanθ too anθ < 11.43 " to 100's es y score y total # of balls te fall of	large)
49 HB5	tan θ							1
			STERS	T			6	4
⁰ i=# of balls ¹ Σ	Scores		³ .985	3	4 6)	⁵ 100	1
6 7	85	8	9		.0		.1	1
.2 .3		.4	.5		16		17	4
.2 .3			.5		10		17	
18 19		20	21		22		23	1
24 25		26	27		28		29	4
			- /					J

15 BALL ROTATION POOL

The game proceeds with a random selecttion of the players' shots being scoring shots. The balls are pocketed in rota-tion (1 through 15). Player skills can be varied by selecting a skill factor between 1 and 20. This determines the relative number of scoring shots to total shots. The random sequences are variable by seed number selection. The program continously tallys each of two players scores.

NOTE:

This program is adapted from HP-65 Users' Library program #03427A by Robert A. Plack.

SOLUTION:

					(after mo	re p	lay
1.2345987		Skill f Seed Initial			11.	R/S *** R/S	
1.	R/S *** R/S	Sunk ba	all #1		12.	*** R/S	
2.	κ≠s *** R∕S	Shoot			13.	***	
3.	* ** R∕S	Shoot Shoot			10.03	GSB2 ***	Re Sc
4.	*** R/S	Shoot			e.	R/S ***	
0.	*** R/S	Miss Plaver	2 shoots		0.	R/S ***	Sr Sł
5.	*** R/S	Shoot			0.	R/S *** R/S	
0.	***	Miss			14.	*** R/S	Sł
4.01	GSB2 ***	Review Score:	Player 1 sunk 4 ba player 2 sunk 1.	alls;	10.05		Ga P1

Guide Lines for Skill Factor Selection:

SKILL FACTOR:

2	То	pocket	15	balls	may	need	120	shots
7	н	"	п	11	11	11	50	н
10	н	п	11		п	п	30	н
13	н	н	п	п	11	н	22	н
15	н	н	н	"	н	11	20	п
17	п	н	п	н	"	п	16	н

SOLUTION:

....

er mo	re p	lay)
	R ∕S	
11.	***	
	R∕S	
12.	***	
	R∕S	
13.	***	
10.03		Review Score
_		Shoot
0.	*** R/S	Shoot
0.	*** R/S	Shoot
0.	***	
	K /S	Shoot

Shoot Game over

layer 1 wins

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEY	S	OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter skill factor (between 1 and 20)	S.F.	ST0	6	
3.	Enter seed (any number)	Seed	ST0	7	
4.	Initialize		GSB	1	-1.00
5.	First player shoots		R/S		Ball # or 0.
6.	A player continues to shoot until a zero		R/S		Ball # or 0.
	is displayed, indicating the shot was missed				
	(If the shot was made, the ball number is				
	displayed). It is then the opponent's turn.				
7.	The score is tallied automatically to				
	review the score.		GSB	2	$S_1S_1 \cdot S_2S_2$
	e.g., 7.02 is read player one has sunk 7				
	balls, player two has sunk 2.				
8.	When all fifteen balls have been sunk, the				
	game is over and the score is displayed.				
9.	For a new game, go to step 4 (or step 2				
	if desired).				
				p	

		<u> </u>				
01 *LBL1	Initia	1170	50 *LBL2			
02 0	Initia	1120	51 RCL2		Get scor	e
03 ST01						
04 ST02						
05 ST08			53 %			
0 6 1	Pointe	r	54 RCL1			
07 STC0	Fornce	I	55 +			
08 CHS			56 FIX2			
09 ST04			57 R/S		Display	score in
10 R/S			58 GTOØ		format	$S_1S_1 \cdot S_2S_2$
10 R/3	"Shoot	11	59 R/S			
12 FIX0						
13 RCL7	Seed					
14 Pi						
15 +						
16 X2						
17 FRC						
18 ST07						
19 EEX						
20 3						
21 ×						
22 FRC						
23 EEX						
24 2						
25 ×						
25 A		- .				
20 KULO 27 ÷	Skill	factor				
27 - 28 INT						
29 6		~~				
30 X4Y?	RND >	b?				
31 GT09	Miss					
32 1		he shot;				
33 ST+i		ent score and				
34 ST+8	coun	τ				
35 1						
36 5						
37 RCL8						
38 X=Y?						
39 GTO2	Game o					
40 R/S	Displa	y ball no.				
41 GT00						
42 *LBL9						
43 RCL4						
44 CHS						
45 ST04		•				
46 ST+0	Switch	pointer				
47 0	D 1					
48 R/S	Displa					
49 GTO0	indi	cate a miss				
	Score	Score REGIS	STERS			
⁰ i,pointer	¹ (playerl)		3	4 <u>±</u>] 5	
	7	8	9	.0	.1	
⁶ Skill factor	<u>Seed</u>	Counter	.5			
.2		.4	.5	16	17	
18	19	20	21	22	23	
24	25	26	27	28	29	
			. ,			

ROULETTE

SOLUTION:

GSB5

R/S

R/S

Seed

Bankroll

#

#

#

0.00 ST02

0.00 STO7 500.27 GSB1

0.28 ***

-500.00 ***

500.27 GSB1

-1000.00 ***

500.27 GSB1

0.14 *** R/S

0.02 ***

The player bets by entering the dollar amount of the bet and the number on which the bet is placed in the form B.##. For instance, 50 m would be entered as 5.07 and 50 m would be 50.27.

A winning number bet pays off at 32-to-1. A winning even-odd bet pays off at one-to-one.

In the "win" sequence, the player's total bankroll is displayed. In the "lose" sequence, the Roulette number is displayed, after which the total bankroll may be displayed by pressing R/S.

NOTES:

NOTES.	-1500.00 ***	
 Bet only whole dollars. The maximum bet is \$99,999,999 	500.27 GSB1 14500.00 ***	Winner
 If your winnings cause your bankroll to exceed 9.9999999 x 10⁹⁹, ERROR will be displayed. 	500.27 GSB1 30500.00 ***	Another winner
This program is adapted from HP-65 Users' Library program #03076A by William A. Sholar.	GSB2 0.25 *** R/S 0.00 ***	Bet even # Total bankroll

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1.	Key in the program			
2.	Initialize		GSB 5	
3.	Store your bankroll	\$	STO 2	
4.	Store seed (any number)	Seed	ST0 7	
5.	Play one of the following:			
5a.	Enter bet and number in BB.## format	B.##	GSB 1	*
	and spin			
5b.	Enter bet on even number and spin	В.	GSB 2	*
5c.	Enter bet on odd number	В.	GSB 3	*
6.	Repeat step 4 as often as desired, or			
	go to step 3 to change your luck.			
	* If you win, your updated total is			
	displayed. If you lose, the spin is			
	displayed. To see your new total.		R/S	Total

	i	U				<u> </u>	1
01 *LBL5	Initi	alize	50	R∕S		Displ	ay total
02 FIX2				*LBL9		-	
03 CLRG			52	RCL7		Seed	
04 *LBL1			53	Pi			
05 INT		-	54	+			
06 ST01	Bet -	> R1	55	χ2			
07 LSTX			56				
08 FRC	#	х D.	57			New s	eed
09 ST03	# Spin	·> K3	58	EEX			
10 GSB9	Shim		59	2			
11 RCL3			60	х			
12 X≠Y?	You 1	ose	61				
13 GT00	You w		62				
14 RCL1 15 3			63				
16 2			64				
16 2 17 X			65				c.
18 ST+2			66			Spin	> 6
19 RCL2			67				
20 R/S	Dicol	ay total	68	R∕S			
21 *LBL2	Bet (even)					
22 0		Creny					
23 ST00							
24 GT08							
25 *LBL3	Bet (odd)					
26 .							
27 5							
28 STO0							
29 *LBL8							
30 R↓	Bet,	\$					
31 ST01							
32 GSB9	Spin						
33 5							
34 0							
35 ×			1				
36 FRC	0 or	.5					
37 RCL0							
38 X≠Y?							
39 GTOØ	V 1						
40 RCL1	You T You v						
41 ST+2		¥ I I I					
42 RCL2							
43 R/S		ay total					
44 *LBL0	Lose	routine					
45 RCL1	Deduc	ct bet					
46 ST-2							
47 RCL6 48 R/S	Disp	lay spin ("pau	ı ş e" ma	ay repla	ce "R/S	")	
48 R/S 49 RCL2				•			
• • 7 KULZ		REG	ISTERS				
⁰ 0 or .5	¹ Bet, \$	² Total	3	#	4		5
6	7 Seed	8	9		.0		.1
Spin	.3	.4	.5		16		17
18	19	20	21		22		23
24	25	26	27		28		29

This program plays tic-tac-toe with the user. The keyboard of the machine is used as the playing board, with each digit representing one of the nine positions, as shown at right. 7 8 q The machine moves first, into a side position (position 2). 4 5 6 The user may move into any of 2 3 1 the eight remaining positions. As play continues, user may move into any unoccupied position for each move.

Tic-tac-toe can be won only if one player makes a mistake. This program takes advantage of user mistakes by completing a row of three, or by setting a trap to force a win. If all user moves are correct, a draw results. The side opening by the machine gives the user a better chance to avoid losing.

This program operates on a game tree look-up basis; a different register, containing the machine responses, may be selected for each of the eight possible user first moves.

NOTES:

- Illegal moves (to occupied positions) gives erroneous results.
- No win, lose, or draw signals are given; the user must keep track of the progress of the game.

This program is adapted from HP-65 Users' Library program #03363A by Delmer D. Hinrichs.


REFERENCE:

Gardner, Martin,<u>Mathematical Puzzles &</u> <u>Diversions</u>, Simon and Schuster, New York, 1959 pages 37-46. EXAMPLE:

Machine plays "X", User plays "O"

		0		0	Х
Х		X		Х	
Turn:	1	2		3	

0	Х	0	х		0	х
0		0			0	0
х		X	Х		Х	Х
4		5			6	



Machine Wins

SOLUTIONS:

1.5873649 ST01 3.5891467 ST02 4.13598 ST03 5.1374698 ST04 6.31578 ST05 7.13589 ST06 0.154763657 ST07 9.31587 ST08	Store constants	(3)	2. 1. 5. 8. 7. 3. 9.	SB1 *** R/S *** R/S *** R/S *** R/S *** Machine Wins
(1) FIX0				
GSB1 2. *** 8. R/S 9. *** 5. R/S 3. *** 6. R/S 1. ***	Machine's lst move Player's lst move Machine Wins	(4)	2. x 8. K 9. x 1. K 3. x 6. K 5. x 4. K	5B1 *** R/S *** R/S *** R/S ***
(2) GSB1			7. *	ua∉. Ma chine wins

2. *** 9. R/S 3. *** 1. R/S

5. 8. ***

R∕S

7. *** Machine wins

STEP	INSTRUCTIONS		KEYS	OUTPUT DATA/UNITS
1.	Enter program			
2.	Store constants:			
	1.5873649		ST0 1	
	3.5891467		ST0 2	
	4.13598		ST0 3	
	5.1374698		STO 4	
	6.31578		ST0 5	
	7.13589		STO 6	
	1.5476365774 [EEX] [CHS] [1]		ST0 7	
	9.31587		ST0 8	
3.	Set display		f Fix	
			0	
4.	Start game	Disvenia	GSB 1	2
5.	Enter move (any number 1-9)	Player's Move	R/S	
6.	Repeat step 5 until the outcome is resolved			
	(HP-29C/19C <u>wins</u> or a draw). The player			
	must keep track of the progress of the			
	game.			
7.	For a new game, go to step 4.			

		-			-
01 *LBL1 02 EEX 03 5 04 ST09 05 2 06 R/S 07 8 08 ST00 09 X=Y? 10 GT06 11 R4 12 *LBL3 13 RCL; 14 GSB0 15 DSZ 16 GT03 17 *LBL0 18 INT 19 X=Y? 20 GT04 21 R4 22 RTN 23 *LBL4 24 RCL; 25 + 26 *LBL5 27 GSB9 28 X=Y? 29 GSB9 30 R/S 31 GT05 32 *LBL9 33 LSTX 34 FRC 35 1 36 0 77 ×	Player's i=8 Is it 8? Yes No Find reg contains respon Match? Hit Miss Save in Find res Is it co No, get time	s lst move lst move . which correct ses last x ponse prrect?	50 3 51 R/S 52 1 53 X=Y? 54 6 55 R/S 56 *LBL8 57 1 58 R/S 59 3 60 X=Y? 61 5 62 R/S 63 *LBL7 64 CLX 65 1 66 X=Y? 67 RCL9 68 RCL7 69 X 70 + 71 GT05 72 R/S		No,response is 3 Enter player's 3rd move Is it 1? Yes, 6 is response No, 1 is response; end game Response is 1 Enter player's move Is it 3? Yes, response is 5 No, response is 3; end game Is it 1? Yes, use last 5 digits in R7 Mult. R7 by 1 or 10 ⁵ same in lastx Go find responses
36 0					
37 X	Cavor a	in last x			
38 INT 39 RTN	Saves ~	τη ταδί λ			
40 *LBL6	Interac				
41 9		egy for 8 s 2nd move			
42 R/S	Player's				
43 4 44 X>Y?	Is it l	or 3?			
45 GT07	Yes				
46 CLX					
47 6	Is it 6	or 7?			
48 X≟Y? 49 GT08		5, 7,			
	L	REGIS		A	
^o Pointer	¹ Library ²	LID.	³ Lib.	⁴ Lib.	⁵ Lib.
⁶ Lib.	7 Lib. 8	Lib.	9 10 ⁵	.0	.1
.2	.3 .4		.5	16	17
18	19 20		21	22	23
24	25 26		27	28	29


Contrary to popular belief, the "Auto-Bell" and "Bell Fruit" brands of slot machines are not rigged. However, the odds at getting a jackpot pattern are extremely low. For example, on a real slot machine, each 'wheel' contains 20 symbols, only one of which is a bar. Thus, with 3-wheels, a 3-bar combination (or 'Jackpot') comes up once every 8000 plays!

This program is more sporting.(depending on the seed used in initialization, this program can be down right generous). When GSB 1 is pressed the 'wheels' spin and a 3 digit decimal is arrived at. (The no. is to the right of the decimal point, ignore the '0' to the left). A dime is deducted from the 'pot' (R_1) . If you win, the payoff amount is paid into the 'pot'. This may be reviewed at any time by pressing RCL 1. Any 3of-a-kind (except cherries) wins \$1.00. Any 2-of-a-kind (except cherries) followed by a 'bar', wins \$1.00. A cherry in the first position wins 20¢. A cherry in the second position, when following the 1st cherry, wins an additional 30¢. All other combinations are "Fruit-Salad" and win you zilch! Good Luck.

NOTES:

It's best to key in a many digit decimal as a seed, as opposed to a small number (use a number like '251.0637948' instead of '3'). Due to the nature of the program, one is generally assured of winning 20¢ (one cherry) immediately following a jackpot (3-bars).

This program is adapted from HP-65 Users' Library program #03044B by Craig A. Pearce.



SOLUTION:

32147.000	GSB5	Seed
	GSB1	Play
0.174	***	\$.20 winner
	GSB2	Pot now has \$.10
0.10	***	
		Play
0.994	***	No luck
		As expected, pot even
0.00	東東東	
	GSB1	
0.067	***	Lost
	GSB1	
0.385	東東東	Lost
	GSR1	
0.905	東京東	Lost
		Sure enough, \$.30 in hole
-0.30	***	
	GSB1	
0.997	***	Lost
	GSB1	
0.120	***	Won \$.20
	GSB1	
0.496	***	Lost
	GSB1	
0.722	***	Lost
	GSB1	
0.999	***	Won \$1.00!
	esb2	
0.40	***	we're ahead

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1.	Key in the program			
2.	Enter seed (any number below 10 ⁶)	Seed	GSB 5	
	and initialize			
3.	Play		GSB 1	0.XYZ
	Winning combinations:			
	0.1nm pays 20¢			
	0.11n pays 30¢			
	0.XXX pays \$1.00			
	0.XXO pays \$1.00			
	(where $X \ge 2$)			
	0.000 pays \$10.00			
4.	Recall pot of winnings (or losses) at		GSB 2	Pot
	any time			
5.	Repeat steps 3-4 any number of times			
6.	For a new game, go to step 2			
7.	For a new seed (at any time)	Seed	STO 8	

24	25	26	27	28	29
18	19	20	21	22	23
.2	.3	.4		16	17
_		Seed	9	.0	.1
<mark>0 Flag</mark> 6	1 Pot	² "X" digit	³ "Y" digit	⁴ "Z" digi	
		REGI	STERS	1.	
49 DSZ	othe for	rwise input a a jackpot	9		
48 9	matc	h? No, input a	1		
47 GT07	Was	no. a 3-way			
46 X≠0?		ot a winner			
44 K. 45 *LBL8	7=0?	if not, no.			
43 STO0 44 R↓		y match			
42 1		flag to show	T		
41 GSB0		otherwise pay for 3-way matc	1		
40 1		ot, test for			
39 GT08	ma	tch?			
38 X≠Y?	Do 2	nd two digits			
36 6107 37 RCL4	Not	a winner			
35 XF17 36 GT07	ma	tch?			
34 RCL3 35 X≠Y?		st two digits			
33 GTO6			82 R/S		
32 X=Y?	A ch	erry?	81 FIX2		
31 RCL2			80 RCL1		
30 1			79 *LBL2		RCL Pot
29 STO4	Z		78 RTN		
28 GSB9			77 INT		
27 -	'		76 ENT†		
26 ST03	Y		75 ×		
25 GSB9			74 0		
23 3762			72 * LBL9 73 1		
22 6383 23 STO2	x		71 GT07 72 *LBL9		Peel off digits
21 5105 22 GSB9			70 GSB0 71 GT07		cherry
20 FRC 21 ST05			69 3 70 CCD0		cherry
19 - 20 FRC			68 .		Pay 30¢ for 2nd
18 LSTX		、	67 GT07		
17 +		(0.XYZ)	66 X≠Y?		Second cherry?
16 6	Δ	djust format	65 RCL3		• • • •
15 EEX			64 GSB0		cherry
14 ST08	RND		63 2		Pay 20¢ for 1
13 ABS	DUD		62 .		
12 COS			61 *LBL6	(Cherry routin
10 0 11 X			60 RTN		
10 3			58 ST+1		
00 KCE0 09 EEX			57 #LBL0		
08 RCL8			56 CHS 57 *LBL0		
07 FIX3			55 1 56 CHS		-104
05 R/S 06 *LBL1			54 . 55 1		-10¢
04 DEG 05 R/S			53 RCL5		
03 ST08			52 *LBL7		
02 CLRG			51 GSB0		
		& initialize	50 1		

BLACKJACK (21)

The player places a bet and draws a card. Jack, Queens, and Kings count as 10 and Aces count as 1. The player keeps taking "hits" (drawing cards) until satisfied with the total. If his total exceeds 21 he loses. The machine draws until its total exceeds the player's.

If its total exceeds 21, you win.

If you win, your bet is added to your bankroll and should you lose, your bet is deducted.

The program uses a random number generator to pick the cards where the probabilitites of drawing a deuce through an ace are all equal.

NOTE:

This program is based on three HP-65 Users' Library programs: 237A by Duke Castle, 1296A by Gary D. Campbell, and 2024A by Mordecai Schwartz, M.D.

SOLUTION:

2.654 ST01 Enter seed

- 50.00 STO2 Enter your bankroll
- 15.00 GSB1 Enter your bet and draw
 - 8. *** Total count after 1st card R/S Hit
 - 18. *** Total after second card
 Stick
- GSE2 Machine plays After each card, the following count totals are flashed: 7,8,9,14, 24 65.00 *** You win, bankroll is now
- \$65.00
- 15.00 GSB1 Bet \$15.00 on another game
 - 1. *** Total count after 1 card R/S

 - 12. *** After 3 cards
 - R/S 14. *** After 4 cards
 - R/S After 5 cards your total of 24 is flashed.
- 50.00 *** You lose-Bankroll is back to \$50.00
- 20.00 GSE1 Now you bet \$20.00
 - 3. *** R/S
 - 5. *****
 - R∕S
 - 10. ***
 - R/S 14. ***
 - R/S
 - [°] 5 card total is flashed:22
 - 30.00 *** Loss reduces bankroll to
- \$30.00 10.00 GSE1 You bet only \$10.00
- 10. ***
- R/S
- 11. *** R/S
- 15. The Decide to stick with a 15 total GSB2 Machine wins again with
 - totals of 10,16 and 20
- 20.00 *** You are left with \$20.00

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1.	Key in the program			
2.	Initialize:			
	Store seed	Any number	ST0 1	
	Store bankroll to play with	WO	ST0 2	
3.	Enter bet and draw a card	Bet	GSB 1	P _l , value of first card
4.	Take a "hit"		R/S	P _i , players total after ith card
5.	Repeat step 4 until satisfied with the			
	total (go to step 6)			
6.	"Stick". Let machine draw,machine draws		GSB 2	M ₁ ,M ₂ W
	until it's total:			
	a) exceeds 21 in which case you win and			
	your bet is added to your bankroll.			
	b) is greater than your total but less			
	than 21, in which case you lose and			
	your bet is subtracted from your			
	bankroll, W.			
7.	For another hand, go to step 3.			

#1 *LB11 Bet 50 PCL1 #2 \$T06 Bet 51 Pi #3 0 52 + 53 yz #4 \$T00 54 FRC 55 yz yz #5 \$T04 55 \$T01 56 1 yz #6 \$ST05 56 1 57 3 yz yz							
27PSE28X>Y?You win29GT0930RCL5Your total31X#Y32X>Y?33GT08You lose34GT02Machine draws35*LBL8again361Set flag37ST0038*LBL939RCL640DS241CHS42CHS43ST+244FIX245RCL245RCL245RCL246R/S47*LBL04819	02 ST06 03 0 04 ST00 05 ST04 06 ST05 07 FIX0 08 *LBL7 09 GSB0 10 ST+5 11 2 12 1 13 RCL5 14 PSE 15 X=Y? 16 GT09 17 X>Y? 18 GT08 19 R/S 20 GT07 21 *LBL2 22 GSB0 23 ST+4 24 2 25 1	Get c. Total You w You 1	in ose	51 Pi 52 + 53 X ² 54 FRC 55 STO1 56 1 57 3 58 × 59 1 60 + 61 INT 62 X>Y? 63 X ² Y 64 RTN		10.	for jack-king
	27 PSE 28 X>Y? 29 GT09 30 RCL5 31 X#Y 32 X>Y? 33 GT08 34 GT02 35 *LBL8 36 1 37 ST00 38 *LBL9 39 RCL6 40 DSZ 41 CHS 42 CHS 43 ST+2 44 FIX2 45 RCL2 46 R/S 47 *LBL0	Your You 1 Machi aga Set f Bet Add w sub Displ (ba ran	total ose ne draws in lag vinnings and tract losses ay total nkroll) dom number				
REGISTERS		A	REGI	STERS			
	0 i	¹ Seed			4 Mach.	Tot	5 Player's To
6 Bet 7 8 9 .0 .1	G			9	.0		.1
.2 .3 .4 .5 16 17		.3	.4	.5	16		17
18 19 20 21 22 23		19	20	21	22		
24 25 26 27 28 29	24	20	20	21	28		29

The program completely simulates the classical cannibal-missionary river crossing problem in the following form:

3 missionaires and 3 partially-civilized cannibals must cross a river with a boat that can hold no more than 3 passengers. At no time may cannibals outnumber missionaries at either bank or on the boat lest the cannibals regress to an earlier mode of behavior! Further, cannibal(s) left aboard the boat alone will run off with it after launching.

Missionaries, cannibals and boat are all initially on the left bank. Press GSB 1 once for each cannibal you wish to put aboard and GSB 2 for each missionary. After each crossing (GSB 3) or return (GSB 4) the right bank distribution (those already across) is displayed in the form $0.C_RM_R$:

For	example:	0.00	No one has crossed-
			initial condition
		0.23	2 cannibals & 3
			missionaries on
			Rt. bank
		0.33	Successful simulation-
			everyone across

After a crossing (GSB 3 or GSB 4), improper operations are appropriately punished:

		DISPLAY
Α.	Impossible crossing- boat on wrong bank	"Error"
Β.	Boat adrift (no one on) or stolen (no miss. on)	1.00
С.	C's outnumber M's on boat	2.00
D.	Boat sinks - 4 or more aboard	4.00
Ε.	M's outnumbered at either bank	3.00

NOTE:

This program is adapted from HP-65 Users' Library program number 02286A by Mordecai Schwartz, M.D.

SOLUTION:

6585 6581 6582 6582 6582 6583 4.00 ***	Initialize Cannibal boards Missionary boards Missionary boards Missionary boards Left to right crossing Boat overloaded-try again
GSB5 GSB2 GSB2 GSB2 GSB3	Initialize M M M > O cannibals & 3 Miss.
0.03 *** GSB2 GSB2 GSB4 3.00 ***	M on right bank M <(right to left C's outnumber M's on left bank
GSB5 GSB1 GSB2 GSB3 0.11 ***	Initialize C M > IC + 1M on right bank M
GSB2 GSB4 0.10 *** GSB1 GSB1 GSB2	< IC + OM on right bank C C M
GSB3 2.00 ***	> Missionaries out- numbered on boat

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KE	YS	OUTPUT DATA/UNITS
1.	Key in the program				
2.	Initialize: Cannibals, missionaries, and		GSB	5	0.00
	boat at left bank				
3.	Load the boat:				
3a.	A cannibal boards		GSB	1	1.00
3b.	A missionary boards		GSB	2	1.00
4.	Repeat 3a/3b until all passengers are				
	loaded				
5.	Cross the river (in the proper direction):				
5a.	Left to right>		GSB	3	0.C _R M _R *
5b.	Right to left <i>≪</i>		GSB	4	0.C _R M _R *
	Output = 0.C _R M _R				
	e.g. 0.23 means 2 cannibals and 3				, .
	missionaries on right bank				
6.	Repeat steps 3-5 until everyone is on				
	right bank.				
7.	For a new game or after an improper				
	operation *, go to step 2.				
	* Outputs after an improper operation:				
	a. Impossible crossing-boat on wrong sid	le			Error
	b. Boat adrift (no one on) or stolen				1.00
	(only C's on)				
	c. M's outnumbered on boat				2.00
	d. Boat overloaded - 4 or more aboard				4.00
	e. C's outnumber M's on either bank or				3.00
	too many M's called, e.g., 1 M on				
	bank and 2 M's loaded aboard				

If crossing involves multiple errors, the display hierarchy is as above.

P		<u> </u>				
01 *LBL5		alize	50 *LBL7		Safe (crossing
02 FIX2			51 0		con	figuration
03 CLRG	R/S		52 ST03			
04 0			53 ST04			
0 5 R/S	1	wital baawda	54 RCL1			
06 *LBL1	A Car	nibal boards	55 EEX			
87 1			56 1			
08 ST+3			57 ÷			
09 R/S			58 RCL2			
10 *LBL2		sionary	59 EEX			
11 1	bo	bards	60 2		Set	t up display 🛛
12 ST+4			61 ÷			
13 R/S			62 +		· · ·	
14 *LBL3	Cross	s river (+)	63 R/S		Displa	ay O.C _R M _R
15 RCL5			64 *LBL4		Cross	river (()
16 X≠0?	left	bank?	65 RCL5			
17 GT00		burne,	66 X=0?		Right	bank?
18 1			67 GTO0		Ĵ	
19 ST05	Set 1	flag	68 0		Clear	flag
20 RCL3		5	69 ST05			-
21 ST+1	Ne	ew right bank	70 RCL3		1	
22 RCL4		istribution	71 ST-1		No	u wight bank
23 ST+2	J		72 RCL4			w right bank stribution
24 *LBL9	Dive		73 ST-2		וס (Stribución
25 1		r crossing er or back	74 GT09			
26 STO0		er or back	75 *LBL0			
28 3100 27 RCL4			76 0		Disnl	ay "Error"
28 X=0?			77 ÷		ызри	
28 X-67 29 GT08			78 *LBL8			
30 ISZ	i = 1		79 RCL0			
			80 R/S		Displa	ay error code
31 RCL3						
32 X>Y?		_				
33 GT08	i = 2	2				
34 +						
35 4						
36 STO0						
37 X≟Y?	i = 4	n				
38 GT08		+				
39 RCL2						
40 X=0?	Safe					
41 GT07						
42 3						
43 X=Y?	Safe					
44 GT07						
45 RCL1						
46 RCL2						
47 DSZ						
48 X≠Y?	i = 3	3				
49 GT08	I					
OEnnon Cada	1 (Diaht bar		STERS	4 MR		5 Flag =set
^o Error Code		^{2 M} Right bank			oat	I lay n=clear
6	7	8	9	.0		.1 o crear
.2	.3	.4	.5	16		17
18	19	20	21	22		23
24	25	26	27	28		29

Using your destroyer, you try to locate the position of the enemy submarine in a 10×10 grid, and then destroy it with a depth charge.

You input a seed (1-100) and the calculator will position the submarine in the center of one of the 100 squares (R,C), where R = row and C = column, and where R and C can each be 0, 1, 2, ...,9.

You make guesses as to where you think the submarine is hiding by taking sonar readings. Input the location of your destroyer (R,C) and press GSB 2. If the submarine is in one of the 8 adjacent squares (or direclty under your destroyer), the calculator will display "1." Otherwise, a "0" will be shown.

When you think you've located the submarine, move your destroyer directly over it (move to the same square) and drop a depth charge. A blinking "l" indicates a hit, while a "O" shows a miss. If you miss, the submarine will move randomly to one of the 4 adjacent squares in the same row or column (but it will not move off the grid).

A depth charge has a range of 0.8. When you position your destroyer for a depth charge drop, (or when you prepare for a sonar reading), you may move any-where on the board, not just to the center of a square. For instance, a depth charge dropped from a (2.5, 6.5) location would destroy any submarine in the center of square (2,6)(2,7)(3,6) and (3,7). Note that sonar's range=1.8

NOTE:

This program is adapted from HP-65 Users' Library program number 01957A by Moshi M. Breiner. SOLUTION:

17.20 GSB1 2.00 ENT†	Seed
2.00 ENTT 2.00 GSB2	Search
8.00 ***	
4.50 ENT†	
4.50 GSB2	Search
0.00 ***	
1.50 ENT†	
6.50 GSB2	Search
1.00 ***	Within 1.8 units
2.00 ENT†	
7.00 GSB2	Search
0.00 ***	
1.00 ENT†	
5.00 GSB2	Search
0.00 ***	
3.00 ENT†	
5.00 GSB2	Search
1.00 ***	Within 1.8 units
2.50 ENT†	
5.50 GSB3	Fire depth charge
8.88 ***	Miss
2.50 ENT†	
3.50 GSB2	Search
1.00 ***	Within 1.8 units
1.50 ENT†	
3.50 GSB3	(61 1, i.e. a) ;+1
1.00 ***	(flashing) Hit!

	0	1	2	3	4	5	6	7	8	9		0	1	2	3	4	5	6	7	8	9	
9											9											9
8											8											8
7											7											7
6											6											6
5											5											5
4											4											4
3											3											3
2											2											2
1											1											1
0											0											0
	0	1	2	3	4	5	6	7	8	9		0	1	2	3	4	5	6	7	8	9	
9											9											9
8											8											8
7											7											7
6											6											6
5											5											5
4											4											4
3											3											3
2											2											2
1											1											1
0											0											0
	0	1	2	3	4	5	6	7	8	9		0	1	2	3	4	5	6	7	8	9	
9											9											9
8											8											8
7											7											7
6											6											6
5											5											5
4											4											4
3											3											3
2											2											2
1											1											1
0											0											0
	0	1	2	3	4	5	6	7	8	9		0	1	2	3	4	5	6	7	8	9	

Playing board for Hunt a Moving Submarine. You might wish to use copies of this page for your games.

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1.	Key in the program			
2.	Initialize and enter seed (any number			
	between 1 and 100)	Seed	GSB 1	1.00
3.	Take a sonar reading	R*	ENT↑	
	or	C*	GSB 2	0 or 1
	Fire a depth charge	R	ENT↑	
		С	GSB 3	0 or
				blinking l
4.	Repeat step 3 until the submarine is			
	destroyed (blinking l.)			
5.	For a new game, go to step 2.			
	* R = Row			
	C = Column			

	01 *LBL1		Initia	alize	49				
	02 CLRG				50				
	03 STO2		Seed		51				
	04 DEG				52				
	05 FIX2				53	÷			ositive if
	06 GSB0				54	DSZ		RND	$\geq 5 \text{ neg if RND}$
	07 ST04		Cs		55	RCL4			3 results
	08 GSB0				56	+		R _s +1	or C _s +1
	09 STO1		R _S		57			If P.	+x = 0 or
	10 1				58				+x = 10, the
	11 ST05		Clear	flag l	59				w coordinate is
	12 R/S				60				-x =
	13 *LB L3		Charge	e fired	61	-			
	14 0				62				P _i +x-9 -9
	15 ST05		Set fl	ag 1				÷ _ 7	A
	15 5785 16 R4				63			i = 1	or 4
	10 K.+ 17 *LBL2				64			E1	,
					65			Flag	1
	18 1 10 CT00					X=0?		7000	
	19 STO0		i = 1		67			Zero	out R₃
	20 R.J				68	1			
	21 RCL4				69	ST05		C1	flag
	22 -		$x_2 - 0$	'S	70	RCL3		clear	flag l
	23 X‡Y				71	R∕ S		Resul	+ I
	24 RCL1				72	*LBL8		Resul	ι I
	25 -		X1 - R	s	73	1			
	26 +P		δ,Eucl	idean dist.	74	PSE		Blink	1.
	27 .				75	GT08			
	28 8					*LBL0			m number
	29 -		δ8	}	77	2		gen	erator
	30 X>0?				78	RCL2			
	31 GTO9		Contin	ue flag l	79	+			
	32 RCL5				80	ST02			
	33 X=0?				81	COS			
	34 GT08		Charge	within	82	ABS			
	35 R4			units	83	EEX			
	36 *LBL 9				84	5			
	37 1				85	x			
	38 STO3				86	FRC			
	39 X <u>4</u> Y?		δ≥1.	8?					
	40 ST-3				87	1			
			Zero o	ut R₃	88	0			
					89	X			
	12 4		4 < RN	ס י	90	INT			6th figure
	13 X4Y?				91	RTN		KND =	6th figure
	14 STO0		1 - 4; U	s will change		R∕S			of cos (R ₂)
	5 RCL1		main	R _s will re- the same	NOT	E: P;	= Any r locat	ow or	column
	ie gsbo		main	che same		1	locat	ion for	r sub.
	17 4								
4	18 .								
				REGI	STERS				0=set
0	i	1 R _S		² Seed	³ Res	ult	4 C _S		⁵ Flag l l=clear
6		7		8	9		.0		.1
		.3		.4	.5		16		17
.2		1		1	1				
.2 18		19		20	21		22		23
.2 18 24		19 25		20	21 27		22 28		23 29

In this game, the gun has a maximum range of 10,000 meters. At the start of a game, the target is at a random direction $(0-360^\circ)$ and random distance (5,000-10,000 meters) from the gun. To locate the target, fire a shell in an arbitrary direction and elevation (i.e., 90° direction and 45° elevation). The display then shows the direction and distance from the shell hit to the target. From this, it is possible to determine where the target was. The target randomly changes its direction from the gun by up to $+5^{\circ}$, and randomly moves towards the gun by 0-1,000 meters after each shell hits. Taking into account the expected movement of the target, a new direction and elevation are estimated, and another shell fired. Corrections are made and shells fired until either the target is blown up (shell hits within 100 meters of target) or the gun is destroyed (target gets within 1,000 meters of gun without being hit).

Note that due to the way that the target moves, the closer it gets to the gun, the easier it is to hit.


A difficulty factor of 10 is stored in register 4. This limits the direction change of the target to $\pm 5^{\circ}$. This may be changed to any real number; for example, to 0 to eliminate zig-zags, or to 20 for a more difficult game. The factor 10 is automatically restored to register 4 at the beginning of the next game.


Independently, the "EEX,3" in steps 62, 63 of the program controls the distance (0-1,000 meters) that the target moves towards the gun after each shell is fired. This may be replaced by, for example, "EEX,2" to make a much easier game. NOTE:

This program is adapted from HP-65 Users' Library program #03320A by Delmer D. Hinrichs.


The range of the gun is given by the formula

Range = Sin 2θ · Maximum Range,





- T = Target
- SH = Shell Hit
- a = Angle, shell hit to target
- b = Distance, shell hit to target
- c = Angle, gun to target
- d = Distance, gun to target
- e = Elevation, angle of gun



SOLUTION: (1)

SOLUTION: (2)

0.00 GSB1 90.00000 ENT† 45.00000 GSB2 -75.14467 ***	Enter seed of O Locate target Target was -75° and 14467 meters from shell impact
-45.00000 ENT† 20.00000 GSB2 141.01832 *** -47.00000 ENT† 12.00000 GSB2 92.00286 *** -46.00000 ENT† 10.00000 GSB2	Shoot You're closer Shoot
-66.00207 *** -47.00000 ENT† 10.00000 GSB2 137.00527 ***	The target is getting further away from your shell
-48.00000 ENT† 7.50000 GSB2 59.00192 *** -47.00000 ENT† 6.50000 GSB2 125.00167 ***	Your shell hit only 167 meters away from target
-46.00000 ENT† 5.00000 GSB2 0.00000 ***	(flashing)-Gun destroyed You lose

63 2	2 is put in step 63 to improve chances of victory
5 .00 000 GSB1	Seed = 5
90.00000 ENT↑ 45.00000 GSB2	Find target
-142.02440 *** 102.74932 ENT1	Target location
30.01001 GSB2 11.00588 ***	Shoot
98.85965 ENT↑ 29.70000 GSB2	Shoot
-176.00229 ***	Missed by 229 m.
100.37360 ENT†	Shoot
29.58000 GSB2 1.00000 ***	(flashing) Target destroyed You win

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEY	rs	OUTPUT DATA/UNITS		
1.	Key in the program						
2.	Enter seed (any number from 0 to 100						
	inclusive) and initialize.	Seed	GSB	1	10.00000		
3.	Enter direction and elevation (both in	Direction	ENT↑				
	degrees) and fire.	Elevation	GSB	2	DDD.MMMMM		
	Display shows direction and distance, shell						
	hit to target, in format: degrees. meters						
4.	Repeat step 3 until:						
4a.	Target is destroyedflashing 1.						
4b.	Gun is destroyedflashing O.						
5.	For a new game, go to step 2.						
]			

et #LEL1 SP X-Y 0 $0 = 1$ Initialize $52 \times X60^\circ$ 0 0 $0 = 1$ Initialize $52 \times X60^\circ$ 0 0 $0 = 1$ Initialize $52 \times X60^\circ$ 0 0 $0 = 1$ Initialize $52 \times X60^\circ$ 0 0 $0 = 1$ Initialize $52 \times X60^\circ$ 0 0 $0 = 1$ Initialize $52 \times X60^\circ$ 0 0 $0 = 1$ Initialize $52 \times X60^\circ$ 0 0 $0 = 1$ Initialize $52 \times X60^\circ$ 0 0 0 $0 = 0$ Starting angle $52 \times X60^\circ$ 0 0 0 0 $10 = 0$ Starting distance $66 \times E20^\circ$ 0 0			0		0	
68 CSEP Starting angle 58 2 69 STO1 59 $\frac{5}{29}$ $\frac{5}{29}$ $\frac{5}{29}$ 11 EEX 61 STr1 New angle 12 3 $\frac{6}{69}$ $\frac{5}{2}$ $\frac{6}{67}$ $\frac{5}{2}$ $\frac{6}{67}$ $\frac{5}{67}$ 14 CSSP $\frac{6}{67}$ $\frac{6}{67}$ $\frac{5}{67}$ $\frac{6}{67}$ $\frac{5}{87}$ $\frac{1000}{77}$ 15 RCL8 $\frac{6}{67}$ $\frac{6}{7}$ $\frac{5}{87}$ $\frac{6}{67}$ $\frac{5}{77}$ $\frac{7}{78}$ $\frac{7}{78}$ $\frac{7}{78}$ $\frac{7}{78}$ $\frac{7}{78}$ $\frac{7}{77}$ $\frac{7}{7$	02 DEG 03 CLRG 04 ST07 05 3 06 6	Initia	alize	51 INT 52 X<0? 53 GSB7 54 ST+3 55 RCL4 56 GSB0	θ	
14 CSB0 1000 15 RCL8 64 CSB0 1000 16 + 66 RCL2 67 EEX New distance 19 0 69 3 69 New distance New distance 20 ST04 27 RCL3 72 RCS Degrees - meters 21 FIX5 Set display 71 RCL3 RCL3 Degrees - meters 22 R/S Elev. Dist. 73 RLBL7 Provember 20 Display blinking 0 25 × 10000 76 FLBL7 77 RCL3 Display blinking 1 26 SIN 79 ST03 88 RL B8 RL B8 B1 B1 <td< th=""><th>08 GSB0 09 ST01 10 5 11 EEX 12 3</th><th>Start</th><th>ing angle</th><th>58 2 59 ÷ 60 – 61 ST+1 62 EEX</th><th>New</th><th>angle</th></td<>	08 GSB0 09 ST01 10 5 11 EEX 12 3	Start	ing angle	58 2 59 ÷ 60 – 61 ST+1 62 EEX	New	angle
181Starting distance68319 θ θ θ θ θ θ 20ST04 21 F1X5Set display 71 RCL3 Pc 22 R/S Z R/S 72 R/S Pc Pc 23 $kEB2$ Elev. Dist. 74 θ Pc Pc Pc 242 74 θ 75 $GT09$ 76 $kEB7$ Pc 25 X 77 $RCL3$ 77 $RCL3$ Pc Pc 26SIN 77 $RCL3$ 77 $RCL3$ Pc Pc 27EEX 10000 76 $kE17$ 77 $RcL3$ Pc 284 10000 76 RH RH RH RH RH 29 X 79 $ST03$ Rd Rd RH RH RH 29 X 79 $ST03$ RH	14 GSB0 15 RCL8 16 +			64 GSB0 65 ST-2 66 RCL2		
21FIX5 22Set display71RCL3 72Degrees \cdot meters23#LBL2 24Elev. Dist.73#LBL8 740Display blinking 025x76FT09 75FT09 75Display blinking 026SIN 2777RCL3 7677RCL3 772841000077RCL3 77RCL3 77Display blinking 026SIN 2978CHS 7979ST03 802841000078RH 81RTN 82B29x79ST03BF29x79ST03BF29x79ST03BF29x80R4 81RTN 82Stle6B30+Rx1y180R4 82Stle6B3308282Stle6BB34ST.085PSE 85BBB35RCL2X2Y285PSE 89BB362-X2-X1Y2-Y189PI 96Radom number generator383-X2-X1Y2-Y189PI 96Radom number 96382-X2-X1Y2-Y189PI 96RA41EEX 4795RX595RN422292FRC93ST07 9444EEX 4995 <td>18 1 19 0</td> <td>Start</td> <td>ing distance</td> <td>68 3 69 X>Y?</td> <td></td> <td></td>	18 1 19 0	Start	ing distance	68 3 69 X>Y?		
$\begin{array}{c c c c c c c c c c c c c c c c c c c $	21 FIX5 Set d ² 22 R/S 23 #LBL2 Elev.			71 RCL3 72 R/S 73 *LBL8		
$\begin{array}{c c c c c c c c c c c c c c c c c c c $	25 × 26 SIN 27 EEX	10000		75 GT09 76 *LBL7 77 RCL3 78 CHS	Dis	play blinking U
34 ST. 0 84 *LBL9 Blink generator 35 RCL1 85 PSE Blink generator 36 RCL2 x_2y_2 87 *LBL0 Random number 38 $\Sigma^ x_2-x_1$ y_2-y_1 89 Pi 39 \overline{x} x_2-x_1 y_2-y_1 89 Pi 40 $\rightarrow P$ $r\theta$ 90 $+$ 91 x^2 42 2 92 FRC 93 stor 93 stor 41 EEX 91 x^2 92 FRC 93 stor 94 x 42 2 93 STO7 94 x 95 RTN 96 x/S 95 RTN 45 R4 r r 96 R/S 96 R/S $a/A \theta \theta$ 5 48 \div 4 95 RTN 96 R/S $a/A \theta$ 5 $a/A \theta$	29 × 30 →R 31 CLΣ 32 Σ+	x ₁ y ₁		80 R↓ 81 RTN 82 *LBL6		
$\begin{array}{c ccccccccccccccccccccccccccccccccccc$	34 ST.0 35 RCL1	XeVe		84 *LBL9 85 PSE 86 GT09		
$\begin{array}{c ccccccccccccccccccccccccccccccccccc$	38 Σ- 39 ⊼ 40 →P 41 EEX	x ₂ -x ₁	y ₂ - y ₁	88 RCL7 89 Pi 90 + 91 X ² 92 FRC		
49 ST03 REGISTERS O 1 Angle 2 Distance 3 Display 4 $\Delta \theta$ 5 0 1 Angle 2 Distance 3 Display 4 $\Delta \theta$ 5 6 7 Seed 8 5000 9 .0 .1 .2 .3 .4 .5 16 17 18 19 20 21 22 23	44 GT06 45 R4 46 EEX 47 5		00?	94 × 95 RTN		
$\begin{array}{c c c c c c c c c c c c c c c c c c c $	49 ST03					
$\begin{array}{c c c c c c c c c c c c c c c c c c c $			REGI	STERS		
6 7 8 9 .0 .1 .2 .3 .4 .5 16 17 18 19 20 21 22 23		I Angle	² Distance	³ Display	<u> </u>	
.2 .3 .4 .5 16 17 18 19 20 21 22 23	6	7	18	9	.0	
	.2	.3	.4	.5	16	17
24 25 26 27 28 29	18	19	20	21	22	23
	24	25	26	27	28	29

NOTES

NOTES

In the Hewlett-Packard tradition of supporting HP programmable calculators with quality software, the following titles have been carefully selected to offer useful solutions to many of the most often encountered problems in your field of interest. These ready-made programs are provided with convenient instructions that will allow flexibility of use and efficient operation. We hope that these Solutions books will save your valuable time. They provide you with a tool that will multiply the power of your HP-19C or HP-29C many times over in the months or years ahead.

Mathematics Solutions Statistics Solutions Financial Solutions Electrical Engineering Solutions Surveying Solutions Games Navigational Solutions Civil Engineering Solutions Mechanical Engineering Solutions Student Engineering Solutions



Reorder No. 00029-14006 Printed in U.S.A. 00029-90010